

Maxima: istruzioni per l'uso

DI ENRICO CENTENARO

enrico.centenaro@istruzione.it
I.I.S. di Piazzola sul Brenta (Pd)

Ver. 0810 (ovvero Ottobre 2008)

Riassunto

Maxima[2] è un Computer Algebra System, un programma che permette di fare calcolo simbolico e numerico, grafici di funzioni in 2 e 3 dimensioni e molto altro ancora.

Queste pagine mostrano una fugace panoramica delle funzionalità di Maxima, avendo in mente un utilizzo fatto in laboratorio, in una scuola secondaria come attività complementare.

Questo documento¹ è *liberamente tratto* da un tutorial del professor Scott Hudson² che mi ha autorizzato a usare il suo elaborato “come meglio credo”. La versione più aggiornata è reperibile all'indirizzo dell'autore[1].

Segnalazioni di errori, suggerimenti e saluti sono molto apprezzati.

1. Quest'opera è stata rilasciata sotto la licenza Creative Commons Attribuzione-Non commerciale-Non opere derivate 2.5 Italia. Visita il sito web <http://creativecommons.org/licenses/publicdomain/> per leggerne una copia o spedisci una lettera a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

2. Professor and Electrical Engineering Coordinator School of Electrical Engineering and Computer Science, Washington State University, Tri-Cities. <http://www.tricity.wsu.edu/~hudson/>

Indice

Indice	2
1 Insiemi	2
2 Logica	4
3 Calcolo Aritmetico e Algebrico	5
4 Definire Espressioni e Funzioni	11
5 Risoluzione di Equazioni	12
6 Limiti	14
7 Derivate	15
8 Integrali	16
9 Vettori e Matrici	18
10 Disegni in 2D e 3D	22
11 Applicazioni	23
11.1 Urti elastici	24
11.2 Urti anelastici	25
11.3 Pallone gonfiato	26
11.4 Studio di funzione	27
11.5 Sommatorie ⁶⁴	28
11.6 Confronto fra grafici ⁶⁷	28
12 Esercizi riassuntivi	28
12.1 Insiemi	28
12.2 Algebra	29
12.3 Analisi	31
Indice analitico	31
Bibliografia	?

1 Insiemi

Con Maxima gli insiemi si possono definire per elencazione attraverso l'istruzione `set`. Basta includere gli elementi che vi sono contenuti. Le operazioni che sono previste sono unione, intersezione, differenza e molte altre (vedi il manuale completo). Qui di seguito riporto alcuni calcoli che dovrebbero chiarire le sintassi dei vari comandi.

```
(%i12) set(); # insieme vuoto
```

64. Questionario n. 2 Sessione ordinaria e suppletiva dell'Esame di Stato 2003-2004.

67. Questionario n. 4 Sessione ordinaria e suppletiva dell'Esame di Stato 2003-2004.

```

(%o12) {}
(%i13) A:set(1,2,3); # insieme composto dai numero 1,2,3
(%o27) {1, 2, 3}
(%i28) B:set(1,3,6,7,8,9,10); # ovvio
(%o21) {1, 3, 6, 7, 8, 9, 10}
(%i22) C:intersection(A,B); # C è l'intersezione fra A e B
(%o22) {1, 3}
(%i23) D:union(A,B); # D è l'unione
(%o23) {1, 2, 3, 6, 7, 8, 9, 10}
(%i24) powerset(C); # Insieme potenza ovvero insieme delle parti
(%o24) {{}, {1}, {1, 3}, {3}}
(%i25) cartesian_product(A,C); # prodotto cartesiano
(%o25) {[1, 1], [1, 3], [2, 1], [2, 3], [3, 1], [3, 3]}
(%i26) setdifference(B,C); # differenza fra B e C
(%o26) {6, 7, 8, 9, 10}
(%i27) cardinality(B); # numero di elementi di B
(%o28) 7

```

Esempio 1. Dati gli insiemi $A = \{1, \dots, 8\}$, $B = \{2, 4, \dots, 20\}$ e $C = \{3, 6, \dots, 18\}$, calcolare $(A \cap C) \cup (B \cap C)$.

```

(%i32) A:set(1,2,3,4,5,6,7,8);
(%o33) {1, 2, 3, 4, 5, 6, 7, 8}
(%i34) B:set(2,4,6,8,10,12,14,16,18,20);
(%o34) {2, 4, 6, 8, 10, 12, 14, 16, 18, 20}
(%i35) C:set(3,6,9,12,15,18);
(%o35) {3, 6, 9, 12, 15, 18}
(%i36) union(intersection(A,C),intersection(B,C));
(%o36) {3, 6, 12, 18}

```

Esempio 2. Verificare la proprietà commutativa della intersezione per gli insiemi A e B . Inoltre utilizzando anche C verificare la proprietà associativa.

```

(%i37) intersect(A,B);
(%o37) {2, 4, 6, 8}
(%i38) intersect(B,A);
(%o38) {2, 4, 6, 8}
(%i39) union(intersection(A,B),C);
(%o39) {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 12, 14, 15, 16, 18, 20}
(%i40) union(A,union(B,C));

```

```
(%o40) {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 12, 14, 15, 16, 18, 20}
```

Esempio 3. Supponendo A l'insieme universo si calcoli $\bar{B}_U$ (complementare di B rispetto a U)

```
(%i41) setdifference(A,B);
```

```
(%o42) {1, 3, 5, 7}
```

Il comando `subset` permette di determinare il sottoinsieme di un insieme composto dagli elementi che rendono vera una certa funzione o predicato (si veda più avanti).

Esempio 4. Si consideri l'insieme $A = \{0, 2, 4, 6, 8, 9, 11\}$ e si ricavi il sottoinsieme B composto dai multipli di 3.

Evidentemente un numero è divisibile per 3 se il resto della divisione per 3 è zero, cioè se il predicato $p(x) = x \text{ div } 3$ è vero. Con Maxima il resto si calcola con l'istruzione `remainder`.

```
(%i1) p(x):=is(remainder(x,3)=0)$ # definisco il predicato con il comando 'is'
```

```
(%i4) p(3); # ovvio
```

```
(%o4) true
```

```
(%i5) p(7); # ovvio
```

```
(%o5) false
```

```
(%i6) A:set(0,2,4,6,8,9,11)$
```

```
(%i7) subset(A,p); # i multipli di 3
```

```
(%o8) {0, 6, 9}
```

```
(%i9) partition_set(A,p); # la partizione di A usando il predicato 'p'
```

```
(%o9) [{2, 4, 8, 11}, {0, 6, 9}]
```

Adesso tocca a te. Aiutandoti con quanto abbiamo fatto fino ad ora fai pratica con Maxima eseguendo gli esercizi seguenti.

Esercizio 1. Sia U l'insieme ambiente composto dai numeri compresi fra 1 e 20 (estremi inclusi). Considerando A l'insieme dei numeri naturali dispari compresi fra 1 e 9 (estremi inclusi), B l'insieme dei numeri naturali compresi fra 5 e 15 (estremi inclusi) e $C = \{2, 4, 5, 11, 17\}$. Calcola:

- a) $(A \cup C) \cap B$
- b) $\bar{A}_U$
- c) $\bar{B}_U$
- d) $A \cap \bar{B}_U$
- e) $\overline{A \cap C}_U$
- f) Insieme delle parti di A
- g) L'insieme di tutti i sottoinsiemi di A aventi 4 elementi.

Esercizio 2. Stabilisci se le seguenti uguaglianze fra gli insiemi dell'esercizio precedente sono vere o false.

- a) $A \cup B = B \cup A$
- b) $(A \cup B) \cap C = A \cup (B \cap C)$
- c) $(A \cup B) \cap C = (A \cap C) \cup (B \cap C)$

2 Logica

Innanzitutto impariamo il nome degli operatori logici che Maxima offre.

Nella tabella che segue, la prima colonna mostra come gli operatori vengono rappresentati in Maxima

`not p` negazione di p
`p and q` congiunzione di p e q
`p or q` disgiunzione di p e q

Come vedete non sono molti, ci auguriamo che il team di sviluppo di Maxima dedichi più attenzione a questo settore.

Facciamo qualche calcolo logico.

```
(%i9) q:true$
(%i10) p:false$
(%i11) p and q;
(%o11) false
(%i12) p or q;
(%o12) true
(%i13) not(p and q) = p or q; # legge di De Morgan
(%o13) true
```

3 Calcolo Aritmetico e Algebrico

Maxima può essere utilizzato come un potente calcolatore.

```
(%i5) 144*17 - 9;
(%o5) 2439
```

Si possono fare dei calcoli con numeri molto grandi, nel calcolo seguente valutiamo la 25-esima potenza di 144:

```
(%i6) 144^25;
(%o6) 910043815000214977332758527534256632492715260325658624
```

Questi calcoli non li fanno le normali calcolatrici. Adesso calcoliamo la radice 25-esima:

```
(%i7) (%o6)^(1/25);
(%o8) 144
```

L'inserimento di espressioni numeriche intere o razionali restituisce il valore semplificato il più possibile come nell'esempio che segue. Per avere i risultati in formato decimale si può usare la funzione `float`. In alternativa, se qualche elemento della riga di input è in formato decimale, allora il risultato sarà visualizzato ancora in decimale. Si noti come vengono utilizzati i nomi delle righe in luogo dei valori stessi.

```
(%i2 102/50+17/11;
(%o2)  $\frac{986}{275}$ 
(%i3) sqrt(%o2);
(%o3)  $\frac{\sqrt{986}}{5\sqrt{11}}$ 
(%i4) float(%o3);
(%o4) 1.893529652647285
```

```
(%i5) sqrt(102/50+17.0/11);
(%o5) 1.893529652647285
```

Oltre alle classiche operazioni Maxima consente di calcolare, il mcm, il MCD, il quoziente e il resto di divisioni intere; permette di verificare se un numero è primo, ne calcola la sua scomposizione e trova i suoi divisori.

Nei calcoli che seguono imparerete a utilizzare queste istruzioni.

```
(%i43) quotient(44,6);
(%o44) 7
(%i45) remainder(44,6);
(%o45) 2
(%i46) quotient(44,6)*6 + remainder(44,6);
(%o46) 44
(%i47) factor(44);
(%o47) 22 11
```

Calcoliamo il mcm e MCD, ricordo che massimo comun divisore in inglese si dice greatest common divisor (**gcd**), mentre il mcm in inglese si chiama least common multiple (**lcm**). Attenzione per utilizzare la funzione lcm bisogna “caricarla” digitando il comando **load** (“functs”);.

```
(%i1) gcd(44,121);
(%o1) 11
(%i2) 44*121/gcd(44,121);
(%o9) 484
(%i10) primep(47);
(%o6) true
(%i7) primep(1234567);
(%o7) false
(%i8) lcm(12,3);
(%o2) 484)
```

Esercizio 3. Calcolare il mcm e MCD fra i seguenti gruppi di numeri:

- a) 3300, 2625
- b) 25875, 16335
- c) 10500, 174150

Esercizio 4. Stabilisci se i seguenti numeri sono primi:
1777, 1234567, 9721

Vediamo qualche operazione con i simboli:

```
(%i9) (x + 2*y)^4;
(%o9) (2 y + x)4
(%i10) expand(%);
(%o10) 16 y4 + 32 x y3 + 24 x2 y2 + 8 x3 y + x4
(%i11) factor(%);
```

(%o11) $(2y + x)^4$

Si noti l'uso del simbolo % per indicare la riga precedente. Come al solito attraverso un esempio introduciamo delle funzioni e mostriamo il loro utilizzo.

Esempio 5. Calcolare il valore numerico delle espressioni con i valori indicati:

$$\frac{(a+b)^2}{a+b} + \left(\frac{1}{a+b} - \frac{1}{a-b} + \frac{2a}{a^2-b^2} \right) \cdot \frac{a^2-b^2}{2}$$

per $a = \frac{1}{2}$ e $b = 1$.

$$\frac{\frac{x(y+x)}{y^3+x^3} + \frac{x-y}{x^2-y^2}}{\left(1 - \frac{x}{y+x}\right)^2} - \frac{1}{x-y}$$

per $y = -\frac{1}{2}$ e $x = 1$

(%i3) `esp:(a+b)^2/(a+b)+(1/(a+b)-1/(a-b)+2*a/(a^2-b^2))*(a^2-b^2)/2;`

(%o3) $\frac{(a^2-b^2)\left(\frac{2a}{a^2-b^2} + \frac{1}{b+a} - \frac{1}{a-b}\right)}{2} + b + a$

(%i4) `subst([a=1/2,b=1],%o3);`

(%o8) 1

(%i9) `esp2:((x*(x+y)/(x^3+y^3)+(x-y)/(x^2-y^2))/((1-x/(x+y))^2))-1/(x-y);`

(%o9) $\frac{\frac{x(y+x)}{y^3+x^3} + \frac{x-y}{x^2-y^2}}{\left(1 - \frac{x}{y+x}\right)^2} - \frac{1}{x-y}$

(%i10) `subst([x=1,y=-1/2],%o9);`

(%o11) $\frac{40}{21}$

Passiamo ora al calcolo letterale: operazioni con i monomi.

Esempio 6. Ridurre alla forma normale il seguente monomio:

$$\left(-\frac{1}{6}\right)x^3\left(+\frac{3}{4}\right)x^2y^3\left(-\frac{2}{5}\right)xy^4z$$

(%i12) `monomio:-1/6*x^3*3/4*x^2*y^3*(-2/5)*x*y^4*z;`

(%o12) $\frac{x^6 y^7 z}{20}$

Esempio 7. Calcolare MCD e mcm fra i seguenti monomi:

a) $168a^2b^4c^3$, $1372a^3b^2$

b) $9625x^2y^3c$, $-94325x^3y^2c^5$

(%i1) `gcd(168*a^2*b^4*c^3,1372*a^3*b^2);`

(%o1) $28a^2b^2$

(%i2) `gcd(9625*x^2*y^3*c,-94325*x^3*y^2*c^2);`

```
(%o2) 1925 c x^2 y^2
(%i3) load("functs"); # serve a caricare, fra l'altro, la funzione lcm=mcm
(%o4) /usr/share/maxima/5.9.2/share/simplification/functs.mac
(%i5) lcm(168*a^2*b^4*c^3,1372*a^3*b^2);
(%o5) 8232 a^3 b^4 c^3
(%i6) lcm(9625*x^2*y^3*c,-94325*x^3*y^2*c^2);
(%o6) -471625 c^2 x^3 y^3
```

Passiamo ora alle operazioni con i polinomi. Utilizzeremo le istruzioni `ratsimp`, `expand`, `factor`, `divide`, `quotient` e `remainder`.

Esempio 8. Esegui le seguenti moltiplicazioni e semplifica quando possibile i monomi simili.

- a) $(a-b)(a+b)(a-2b)(a+3b)$
- b) $(a^2-3b^2)(a+b)^3(a^3-8b^3)$
- c) $(a-2b)(a+2b)(a+b)^2$

```
(%i7) e1:(a-b)*(a+b)*(a-2*b)*(a+3*b);
(%o7) (a-2 b) (a-b) (b+a) (3 b+a)
(%i8) expand(e1);
(%o8) 6 b^4 - a b^3 - 7 a^2 b^2 + a^3 b + a^4
(%i9) e2:(a^2-3*b^2)*(a+b)^3*(a^3-8*b^3);
(%o9) (b+a)^3 (a^2-3 b^2) (a^3-8 b^3)
(%i10) expand(e2);
(%o10) 24 b^8 + 72 a b^7 + 64 a^2 b^6 - 3 a^3 b^5 - 33 a^4 b^4 - 16 a^5 b^3 + 3 a^7 b + a^8
(%i11) e3:(a-2*b)*(a+2*b)*(a+b)^2;
(%o11) (a-2 b) (b+a)^2 (2 b+a)
(%i12) expand(e3);
(%o12) -4 b^4 - 8 a b^3 - 3 a^2 b^2 + 2 a^3 b + a^4
```

Esempio 9. Trovare quoziente e resto delle seguenti divisioni polinomiali.

- a) $(5x^4 - 2x^3 + x^2 - 3x + 7) \div (x + \frac{1}{2})$
- b) $(4x^6 - 2x^5 + x^4 - 12x^2 + 42x^2 - 20x - 69) \div (x + 3)$

```
(%i13) p[1]:5*x^4-2*x^3+x^2-3*x+7;
(%o13) 5 x^4 - 2 x^3 + x^2 - 3 x + 7
(%i14) p[2]:x+1/2;
(%o14) x + 1/2
(%i15) quotient(p[1],p[2]);
(%o15) 40 x^3 - 36 x^2 + 26 x - 37
      8
(%i16) remainder(p[1],p[2]);
```



```
(%o16) 149
16
(%i17) s[1]:4*x^6-2*x^5+x^4-12*x^2+42*x^2-20*x-69;
(%o17) 4x^6 - 2x^5 + x^4 + 30x^2 - 20x - 69
(%i18) s[2]:x+3;
(%o18) x + 3
(%i19) divide(s[1],s[2]);
(%o19) [4x^5 - 14x^4 + 43x^3 - 129x^2 + 417x - 1271, 3744]
```

Esempio 10. Fattorizza i seguenti polinomi.

- a) $2x^4 + 5x^3 - 8x^2 - 17x - 6$
 b) $6x^6 - 29x^5 + 24x^4 + 40x^3 - 36x^2 - 11x + 6$

```
(%i20) factor(2*x^4+5*x^3-8*x^2-17*x-6);
(%o20) (x - 2)(x + 1)(x + 3)(2x + 1)
(%i21) factor(6*x^6-29*x^5+24*x^4+40*x^3-36*x^2-11*x+6);
(%o21) (x - 3)(x - 2)(x - 1)(x + 1)(2x + 1)(3x - 1)
```

Esercizio 5. Esegui le seguenti divisioni:

- a) $(x^2 - x - 12) \div (x - 4)$
 b) $(x^5 + x^2 - x^4 - x) \div (x - 1)$

Esercizio 6. Fattorizza i seguenti polinomi:

- a) $2x^4 + 10x^3 - 50x^2 - 250x$
 b) $3a^2x^3 - 15a^3x^2 - 42a^4x$

Esempio 11. Semplifica le seguenti frazioni algebriche:

$$\frac{x^3 + 3x^2 - x - 3}{x^2 + 2x + 1} \left(\frac{1}{a - b} + \frac{1}{a + b} \right) \cdot \frac{4a - 4b}{ab + b^2} \cdot \frac{a^2 + 2ab + b^2}{4a}$$

```
(%i22) Frac:(x^3+3*x^2-x-3)/(x^2+2*x+1);
(%o22) x^3 + 3x^2 - x - 3
x^2 + 2x + 1
(%i23) expand(Frac);
(%o25) x^3
x^2 + 2x + 1 + 3x^2
x^2 + 2x + 1 - x
x^2 + 2x + 1 - 3
x^2 + 2x + 1
(%i26) factor(Frac);
(%o26) (x - 1)(x + 3)
x + 1
(%i27) ratsimp(Frac);
(%o27) x^2 + 2x - 3
x + 1
```

Talvolta è conveniente utilizzare `fullratsimp` in luogo di `ratsimp` perché vengono eseguite delle semplificazioni “non algebriche”.

```
(%i1) expr: (x^(a/2)+1)^2*(x^(a/2)-1)^2/(x^a-1);
```

```
(%o4) 
$$\frac{\left(x^{\frac{a}{2}}-1\right)^2\left(x^{\frac{a}{2}}+1\right)^2}{x^a-1}$$

```

```
(%i5) ratsimp(expr);
```

```
(%o5) 
$$\frac{x^{2a}-2x^a+1}{x^a-1}$$

```

```
(%i6) fullratsimp(expr);
```

```
(%o6) 
$$x^a-1$$

```

Esercizio 7. Semplifica la seguente frazione algebrica:

$$\left(\frac{1}{a-b} + \frac{1}{a+b}\right) \cdot \frac{4a-4b}{ab+b^2} \cdot \frac{a^2+2ab+b^2}{4a}$$

Successivamente trova il suo valore facendo le seguenti sostituzioni: $a=3, b=2$

Sono possibili altre manipolazioni di espressioni algebriche/trascendenti. Per esempio si può dividere una espressione in un numero predefinito di polinomi, oppure è possibile prendere i primi n monomi di un polinomio dopo che è stato ordinato, oppure è possibile dividere un polinomio in due polinomi: quello che contiene una certa variabile e quello che non la contiene. Gli esempi che seguono dovrebbero essere chiarificatori.

```
(%i7) partition(x^2+x*y+3*y+x-1,x); # x è la variabile che discrimini
```

```
(%o7) 
$$[3y-1, xy+x^2+x]$$

```

```
(%i7) part(x^3-x^2+x-1,2); # prendo il secondo elemento da sinistra
```

```
(%o8) 
$$-x^2$$

```

```
(%i9) part(-x^2+x-1+x^3,2); # idem dopo l'ordinamento (medesimo risultato!)
```

```
(%o9) 
$$-x^2$$

```

```
(%i10) expr: (x+y)/2-sqrt(x-sqrt(x+1))+log(sin(sqrt(x+y)));
```

```
(%o10) 
$$\log\left(\sin\left(\sqrt{y+x}\right)\right) + \frac{y+x}{2} - \sqrt{x-\sqrt{x+1}}$$

```

```
(%i11) pickapart(expr,0);
```

```
(%t11) 
$$\log\left(\sin\left(\sqrt{y+x}\right)\right) + \frac{y+x}{2} - \sqrt{x-\sqrt{x+1}}$$

```

```
(%o11) %t11
```

```
(%i12) pickapart(expr,1);
```

```
(%t12) 
$$\log\left(\sin\left(\sqrt{y+x}\right)\right)$$

```

```
(%t13) 
$$\frac{y+x}{2}$$

```

```
(%t14) 
$$-\sqrt{x-\sqrt{x+1}}$$

```

```
(%o14) %t14 + %t13 + %t12
```

```
(%i14) pickapart(expr,2);
```

(%t15) $\sin(\sqrt{y+x})$

(%t16) $y+x$

(%t17) $\sqrt{x-\sqrt{x+1}}$

(%o17) $-\%t17 + \frac{\%t16}{2} + \log(\%t15)$

(%i17) `pickapart(expr,3);`

(%t18) $\sqrt{y+x}$

(%t19) $x-\sqrt{x+1}$

(%o19) $\frac{y+x}{2} - \sqrt{\%t19} + \log(\sin(\%t18))$

Se l'espressione è complessa utilizzando `realpart` e `imagpart` la si scompone nella sua parte reale e in quella immaginaria.

(%i19) `z: (x+%i*y)^2;`

(%o19) $(iy+x)^2$

(%i20) `[realpart(z),imagpart(z)];`

(%o20) $[x^2-y^2, 2xy]$

4 Definire Espressioni e Funzioni

Quando si eseguono operazioni complesse, che coinvolgono espressioni lunghe e complicate da trascrivere, è certamente utile dare loro un nome cosicché sia possibile utilizzare il nome in luogo delle espressioni. Come abbiamo visto nella sezione precedente Maxima fa questa operazione in modo automatico con ogni riga di input e di output usando `%i(numero)` e `%o(numero)`. Per personalizzare il nome si utilizza la seguente forma (che di fatto abbiamo già usato!):

`NomeEspressione : espressione;`

L'esempio seguente, mostra questo utilizzo, si noti che inserendo semplicemente il nome della espressione si ottiene come risultato l'espressione.

(%i14) `x:102/50+17/11;`

(%o14) $\frac{986}{275}$

(%i15) `y:sqrt(x);`

(%o15) $\frac{\sqrt{986}}{5\sqrt{11}}$

(%i16) `y;`

(%o16) $\frac{\sqrt{986}}{5\sqrt{11}}$

In matematica, scienze, ingegneria e fisica le lettere greche vengono utilizzate con profusione. Maxima riconosce nomi⁶⁸ come `alpha`, `beta`, `rho`, come viene mostrato nell'esempio. Si noti l'uso delle parentesi quadrate per denotare delle variabili indicizzate come in `beta[rho]`.

```
(%i9) beta[z] : sqrt(beta^2-beta[rho]^2);
```

```
(%o9)  $\sqrt{\beta^2 - \beta_p^2}$ 
```

```
(%i10) beta[z];
```

```
(%o10)  $\sqrt{\beta^2 - \beta_p^2}$ 
```

```
(%i11) beta[rho];
```

```
(%o11)  $\beta_p$ 
```

In Maxima le funzioni si definiscono come segue:


```
f(x) := espressione;
```

L'uso viene mostrato qui di seguito. Si noti che le espressioni possono contenere anche variabili indefinite, quando però a queste si assegna un valore, la funzione riflette questa variazione.

```
(%i1) f : exp(-sin(x^2));
```

```
(%o1)  $e^{-\sin(x^2)}$ 
```

```
(%i2) plot2d(f, [x,0,5]);
```



5 Risoluzione di Equazioni

Il comando `solve` risolve, quando possibile, una o più equazioni (sistemi). Negli esempi che seguiranno calcoleremo le soluzioni di equazioni e sistemi.

Esempio 12. Risolvere l'equazione:

$$8\left(\frac{1}{2} - x\right) - 3(2x - 1) = 2(x + 3) - 11$$

68. Si noti che sono scritti in inglese. Altri: epsilon, phi, psi, lambda, ...

```
(%i28) solve(8*(1/2-x)-3*(2*x-1)=2*(x+3)-11);
```

```
(%o28)  $\left[ x = \frac{3}{4} \right]$ 
```

Esempio 13. Risolvere la seguente equazione parametrica rispetto alla variabile x .

$$x(a-3) + \frac{2(x+1)}{a} = 1$$

```
(%i29) eq:x*(a-3)+2*(x+1)/a=1;
```

```
(%o29)  $\frac{2(x+1)}{a} + (a-3)x = 1$ 
```

```
(%i30) ratsimp(eq);
```

```
(%o31)  $\frac{(a^2 - 3a + 2)x + 2}{a} = 1$ 
```

```
(%i32) factor(a^2-3*a+2);
```

```
(%o34)  $(a-2)(a-1)$ 
```

```
(%i35) solve(eq,x);
```

```
(%o35)  $\left[ x = \frac{1}{a-1} \right]$ 
```

Esercizio 8. Risolvere le seguenti equazioni intere:

a) $3(2x+1) - 2(3x+1) = 4x + 3 - 4(x-1)$

b) $\frac{x-4}{5} - \frac{2-x}{4} + \frac{7}{10} \left(\frac{3x-1}{2} + \frac{2(x-3)}{5} \right) = 0$

c) $a(x-1) + ax = x(2a+1) + 1 + 3a$

Vediamo ora un esempio nel quale viene risolto un sistema lineare di due equazioni in due incognite. Useremo il comando `solve` i cui argomenti saranno due liste: la lista delle equazioni e la lista delle incognite. Le liste sono racchiuse da parentesi quadrate [...] e gli elementi sono separati dalla virgola.

Esempio 14. Risolvi il seguente sistema lineare:

$$\begin{cases} 4x - 5y = 3 \\ 2x + 3y = 1 \end{cases}$$

```
(%i36) solve([4*x-5*y=3,2*x+3*y=1],[x,y]);
```

```
(%o36)  $\left[ \left[ x = \frac{7}{11}, y = -\frac{1}{11} \right] \right]$ 
```

In questo modo risolviamo senza conoscere il metodo che Maxima applica. Supponiamo di voler applicare il metodo di Cramer, in questo caso dobbiamo scrivere la matrice dei coefficiente e successivamente mettere il sistema in forma matriciale.

```
(%i37) A:matrix([4,-5],[2,3]);
```

```
(%o3)  $\begin{pmatrix} 4 & -5 \\ 2 & 3 \end{pmatrix}$ 
```

```
(%i4) X:matrix([x[1]],[x[2]]);
```

```
(%o4)  $\begin{pmatrix} x_1 \\ x_2 \end{pmatrix}$ 
```

```
(%i5) B:matrix([3],[1]);
```

```
(%o5)  $\begin{pmatrix} 3 \\ 1 \end{pmatrix}$ 
```

```
(%i6) A.X=B;
```

```
(%o6)  $\begin{pmatrix} 4x_1 - 5x_2 \\ 3x_2 + 2x_1 \end{pmatrix} = \begin{pmatrix} 3 \\ 1 \end{pmatrix}$ 
```

```
(%i7) X=invert(A).B;
```

```
(%o7)  $\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \begin{pmatrix} \frac{7}{11} \\ -\frac{1}{11} \end{pmatrix}$ 
```

E' possibile utilizzare anche una istruzione ottimizzata per sistemi lineari, come viene mostrato nell'esempio.

```
(%i8) linsolve([4*x-5*y=3,2*x+3*y=1],[x,y]);
```

```
(%o8)  $\left[ x = \frac{7}{11}, y = -\frac{1}{11} \right]$ 
```

Esercizio 9. Risolvere i seguenti sistemi lineari:

a) $\begin{cases} 3(x+y) - 6y + 8 - 3x = 2(3x - 2y) \\ 2x + 3(2y - 3x) = 2x + 1 \end{cases}$

b) $\begin{cases} (x+1)^2 - 2y = x^2 - 2x + 3 \\ (x-2)(x+3) + 3x = y + (x-2)^2 \end{cases}$

Nell'esempio viene risolta l'equazione $e^{x^2-3x+2} = 1/2$ rispetto alla variabile x , la riga di output successiva mostra la soluzione esatta. L'uso di `float` restituisce i valori numerici ed evidenzia che le soluzioni sono numeri complessi.

In Maxima la funzione `log` è il logaritmo naturale che in altri ambiti viene indicato con $\ln(x)$ oppure con $\log_e(x)$. Si noti che nonostante l'equazione sia stata nominata con la variabile `eq`, si può sempre utilizzare `%o2`, avremmo potuto utilizzare anche il comando `solve(eq,x);`.

```
(%i2) eq : exp(x^2-3*x+2)=1/2;
```

```
(%o2)  $e^{x^2-3x+2} = \frac{1}{2}$ 
```

```
(%i3) solve(%o2,x);
```

```
(%o3)  $\left[ x = -\frac{\sqrt{1-4\log(2)}-3}{2}, x = \frac{\sqrt{1-4\log(2)}+3}{2} \right]$ 
```

```
(%i4) float(%o3);
```

```
(%o4)  $[x = -0.5(1.33138601548904i - 3.0), x = 0.5(1.33138601548904i + 3.0)]$ 
```

Una istruzione ottimizzata per la risoluzione di sistemi algebrici è `algsys` che sintassi analoga a `solve` e che utilizzeremo più avanti.

6 Limiti

Il calcolo di limiti è molto semplice, basta utilizzare la funzione `limit(f,var,valore[,verso])` dove `f` rappresenta la funzione, `var` la variabile rispetto alla quale si vuole calcolare, `valore` è il valore sul quale si vuole calcolare il limite e, opzionalmente, si può specificare il verso `plus` o `minus` per indicare il limite destro e sinistro.

Esempio 15. Calcolare i seguenti limiti:

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x}$$

$$\lim_{x \rightarrow 0^-} \frac{x}{|x|}$$

$$\lim_{x \rightarrow +\infty} \left(1 + \frac{1}{x}\right)^x$$

```
(%i6) limit(sin(x)/x,x,0);
(%o6) 1
(%i7) limit(x/abs(x),x,0);
(%o7)  $\lim_{x \rightarrow 0} \frac{x}{|x|}$ 
(%i8) limit(x/abs(x),x,0,minu);
(%o8) -1
(%i9) assume(x>0);
(%o9)  $[x > 0]$ 
(%i10) limit(x/abs(x),x,0);
(%o10) 1
(%i11) limit((1+1/x)^x,x,inf);
(%o11) e
(%i12)
```

Nel secondo limite si vede come calcolare il limite destro o sinistro, in alternativa si può utilizzare il comando `assume`. Si noti che se non ci sono le condizioni, il limite non viene calcolato. Il risultato del terzo integrale è la costante di Nepero `e`.

E' possibile combinare anche funzionali diversi, per esempio possiamo calcolare un integrale generalizzato.

Esempio 16. Si calcoli il seguente integrale generalizzato:

$$\int_0^{+\infty} e^{-t^2} dt$$

```
(%i23) f(x):=integrate(%e^(-t^2),t,0,x);
(%o23)  $f(x) := \text{integrate}(e^{-t^2}, t, 0, x)$ 
(%i24) limit(f(x),x,+inf);
(%o26)  $\frac{\sqrt{\pi}}{2}$ 
```

7 Derivate

L'operatore `diff` calcola la derivata simbolica della funzione specificata come primo argomento rispetto alla variabile del secondo.

```
(%i17) g:exp(sin(t^2));
```

```
(%o24)  $e^{\sin(t^2)}$ 
```

```
(%i25) diff(g,t);
```

```
(%o26)  $2t \cos(t^2) e^{\sin(t^2)}$ 
```

Opzionale è il terzo argomento che indica l'ordine della derivata, il valore predefinito è uno (viene calcolata la derivata prima).

```
(%i28) diff(g,t,2);
```

```
(%o28)  $-4t^2 \sin(t^2) e^{\sin(t^2)} + 4t^2 \cos(t^2)^2 e^{\sin(t^2)} + 2 \cos(t^2) e^{\sin(t^2)}$ 
```

E' anche possibile verificare se una data funzione è soluzione di una certa equazione differenziale, come mostra il seguente esempio:

```
(%i30) y:exp(-t^2);
```

```
(%o30)  $e^{-t^2}$ 
```

```
(%i31) diff(y,t,2)+2*t*diff(y,t)+2*y;
```

```
(%o32) 0
```

Che ci permette di concludere che la funzione e^{-t^2} è soluzione della equazione differenziale:

$$y'' + 2xy' + 2y = 0$$

Si può utilizzare `diff` anche annidato tipicamente durante il calcolo di derivate parziali. Nell'esempio che segue calcoliamo $\frac{\partial^2}{\partial t \partial s} \sin(\beta_t t) \cos(\beta_s s) e^{-i\beta_w w}$.

```
(%i33) f:sin(beta[t])*t*cos(beta[s]*s)*exp(-%i*beta[w]*w);
```

```
(%o34)  $\cos(s\beta_s) t \sin(\beta_t) e^{-i w \beta_w}$ 
```

```
(%i35) diff(diff(f,t),s);
```

```
(%o35)  $-\beta_s \sin(s\beta_s) \sin(\beta_t) e^{-i w \beta_w}$ 
```

Esercizio 10. Calcolare la derivata delle seguenti funzioni:

$$y = e^{\sin(x)} \cos(1/x)$$

$$Y = \left(1 + \frac{1}{x}\right)^{\frac{1}{x}}$$

$$y = \frac{\sin(x^2) + \cos(x^2)}{\tan(x^2)}$$

8 Integrali

Il calcolo degli integrali indefiniti (le primitive) può essere eseguito utilizzando l'istruzione `integrate`. Il primo argomento è la funzione della quale si vuole calcolare la primitiva, il secondo argomento è la variabile rispetto alla quale fare il calcolo (la variabile di integrazione). Nell'esempio seguente calcoliamo delle primitive sia direttamente che attraverso la definizione preventiva delle funzioni integrande.

```
(%i37) integrate(t^2*sin(t),t);
```

```
(%o40)  $2t \sin(t) + (2 - t^2) \cos(t)$ 
```



```
(%i41) f:t^2*sin(t);
```

```
(%o41) t^2 sin (t)
```

```
(%i42) integrate(f,t);
```

```
(%o42) 2 t sin (t) + (2 - t^2) cos (t)
```

```
(%i43) integrate(exp(t^3),t);
```

```
(%o44)  $\int e^{t^3} dt$ 
```

E' possibile anche calcolare integrali definiti, basta specificare come terzo e quarto argomento gli estremi di integrazione.

```
(%i45) integrate(f,t,0,%pi);
```

```
(%o45)  $\pi^2 - 4$ 
```

Precedendo il comando integrate da un apostrofo⁶⁹ si fa in modo che Maxima non esegua il calcolo.

```
(%i47) 'integrate(f,t,0,5*pi);
```

```
(%o47)  $\int_0^{5\pi} t^2 \sin(t) dt$ 
```

E' anche possibile cambiare la variabile con il comando `changevar` come si evidenzia dall'esempio seguente.

```
(%i48) changevar(%o47,t^2-s,s,t);
```

```
(%o54)  $\frac{\int_0^{25\pi^2} \sin(\sqrt{s}) \sqrt{s} ds}{2}$ 
```

Il primo argomento rappresenta l'integrale, il secondo il cambiamento di variabile, cioè la equazione che lega le due variabili (si può scrivere $s = t^2$ oppure $t^2 - s$), il terzo la nuova variabile e il quarto la vecchia variabile.

Si possono calcolare anche integrali multipli, basta annidare la funzione integrate. Nell'esempio seguente calcoleremo $\int_0^1 \int_0^y e^{-t} dt dy$. Si noterà che Maxima chiede se y è positivo, negativo o nullo, noi risponderemo `positive`; (non dimenticare il punto e virgola finale).

```
(%i57) integrate(integrate(exp(-s),s,0,w),w,0,1);
```

```
Is w positive, negative, or zero?positive;
```

```
(%o59)  $e^{-1}(e+1) - 1$ 
```

```
(%i60) ratsimp(%);
```

```
(%o60)  $e^{-1}$ 
```

```
(%i61) float(%);
```

```
(%o61) 0.36787944117144
```

Volendo potremmo definire una funzione utilizzando l'integrale $F(t) = \int_0^t \exp(-s) ds$, e verificare il teorema fondamentale del calcolo integrale⁷⁰.

```
(%i62) F(t):='integrate(exp(-s),s,0,t);
```

69. In verità ciò succede anche con tutti gli altri operatori.

70. Si noti l'uso dell'apice (') prima della istruzione integrate che impedisce a Maxima di eseguire l'istruzione.

```
(%o18)  $F(t) := \int_0^t \exp(-s) \, ds$ 
(%i19) diff(F(t),t);
Is t positive, negative, or zero?positive;
(%o68)  $e^{-t}$ 
```

Se vogliamo calcolare un integrale definito approssimato, possiamo utilizzare il comando `romberg`⁷¹ con la stessa sintassi di `integrate`.

```
(%i69) romberg(exp(-t^3),t,0,1);
(%o69) 0.80751119008366
(%i70) f(t):=exp(-t^3)*sin(t);
(%o70)  $f(t) := \exp(-t^3) \sin(t)$ 
(%i71) romberg(f(t),t,0,1);
(%o71) 0.32619406816191
```

Esercizio 11. Calcolare i seguenti integrali:

$$\int \sin(x) \log(x) dx$$

$$\int \frac{x^2 - x + 1}{x^4 + 2x^2 + 1} dx$$

$$\int_0^1 \sin(\cos(x)) dx$$

9 Vettori e Matrici

Generalmente, per calcoli con matrici e vettori numerici è preferibile utilizzare strumenti come Scilab o Octave. Maxima è consigliato nel caso si debbano fare calcoli simbolici o quando si vogliono ottenere risultati esatti.

Il comando `matrix` permette di dichiarare una matrice di arbitraria dimensione, se si vuole costruire la matrice identità si usa il comando `ident(dim)` mettendo al posto di `dim` la dimensione.. Per costruire nuove matrici si possono utilizzare i comandi `addcol` e `addrow` che affiancano o sovrappongono matrici fra loro.

```
(%i21) A: matrix([a,b],[c,d]);
(%o21)  $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$ 
(%i22) B: addcol(A,A);
(%o22)  $\begin{pmatrix} a & b & a & b \\ c & d & c & d \end{pmatrix}$ 
(%i23) C: addrow(A,A);
(%o23)  $\begin{pmatrix} a & b \\ c & d \\ a & b \\ c & d \end{pmatrix}$ 
```

⁷¹. Che, ovviamente, utilizza il metodo di Romberg. Sono previsti anche altri metodi di integrazione numerica per questo si fa riferimento al manuale di Maxima.

Le limitazioni dei comandi `addrow` e `addcol` sono evidenti: posso sovrapporre due matrici se hanno lo stesso numero di colonne, mentre le posso affiancare se hanno lo stesso numero di righe.

Se, viceversa, si vogliono estrarre delle righe o delle colonne da una matrice si usano i comandi `col(index)` e `row`, come mostra il seguente esempio.

```
(%i24) A: matrix([0,1,a],[1,b,0],[c,0,0]);
```

```
(%o24) 
$$\begin{pmatrix} 0 & 1 & a \\ 1 & b & 0 \\ c & 0 & 0 \end{pmatrix}$$

```

```
(%i25) row(A,2);
```

```
(%o25) 
$$(1 \quad b \quad 0)$$

```

```
(%i26) col(A,3);
```

```
(%o27) 
$$\begin{pmatrix} a \\ 0 \\ 0 \end{pmatrix}$$

```

```
(%i28) I: ident(3);
```

```
(%o38) 
$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```

Determinanti e inverse di matrici possono essere calcolati con i comandi `determinant` e `invert` mentre il comando `adjoint` restituisce la matrice aggiunta (cioè la matrice inversa moltiplicata per il determinante).

```
(%i28) determinant(A);
```

```
(%o28) 
$$-a b c$$

```

```
(%i29) invert(A);
```

```
(%o29) 
$$\begin{pmatrix} 0 & 0 & \frac{1}{c} \\ 0 & \frac{1}{b} & -\frac{1}{bc} \\ \frac{1}{a} & -\frac{1}{ab} & \frac{1}{abc} \end{pmatrix}$$

```

```
(%i30) adjoint(A);
```

```
(%o30) 
$$\begin{pmatrix} 0 & 0 & -ab \\ 0 & -ac & a \\ -bc & c & -1 \end{pmatrix}$$

```

```
(%i31) invert(A)*determinant(A);
```

```
(%o31) 
$$\begin{pmatrix} 0 & 0 & -ab \\ 0 & -ac & a \\ -bc & c & -1 \end{pmatrix}$$

```

L'operazione di moltiplicazione (riga/colonna) si esegue col comando `.` mentre per la somma e la sottrazione si usano i soliti simboli.

```
(%i32) B: invert(A);
```

$$(\%o32) \begin{pmatrix} 0 & 0 & \frac{1}{c} \\ 0 & \frac{1}{b} & -\frac{1}{bc} \\ \frac{1}{a} & -\frac{1}{ab} & \frac{1}{abc} \end{pmatrix}$$

(%i33) A.B;

$$(\%o33) \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Attenzione: se si usa il simbolo * fra due matrici o si eleva una matrice a un certo esponente si agisce su ogni singolo elemento.

(%i34) C: matrix([1,2],[2,1]);

$$(\%o34) \begin{pmatrix} 1 & 2 \\ 2 & 1 \end{pmatrix}$$

(%i35) C*2;

$$(\%o35) \begin{pmatrix} 2 & 4 \\ 4 & 2 \end{pmatrix}$$

(%i36) C*C;

$$(\%o36) \begin{pmatrix} 1 & 4 \\ 4 & 1 \end{pmatrix}$$

(%i37) C^(-1);

$$(\%o37) \begin{pmatrix} 1 & \frac{1}{2} \\ \frac{1}{2} & 1 \end{pmatrix}$$

Il polinomio caratteristico di una matrice si ottiene calcolando direttamente $p(\alpha) = \det(A - \alpha I)$, oppure con il comando `charpoly`.

(%i39) expand(determinant(A-alpha*ident(3)));

$$(\%o41) -abc + a\alpha c + \alpha^2 b - \alpha^3 + \alpha$$

(%i42) expand(charpoly(A,alpha));

$$(\%o44) -abc + a\alpha c + \alpha^2 b - \alpha^3 + \alpha$$

Il comando `eigenvalue` di una matrice quadrata restituisce gli autovalori sotto forma di due liste, la prima contiene gli autovalori mentre la seconda le loro molteplicità. Il comando `eigen-vectors` fornisce per ogni autovalore gli autovettori.

(%i45) A: matrix([0,1,0,0],[3*w^2,0,0,2*w],[0,0,0,1],[0,-2*w,0,0]);

$$(\%o45) \begin{pmatrix} 0 & 1 & 0 & 0 \\ 3w^2 & 0 & 0 & 2w \\ 0 & 0 & 0 & 1 \\ 0 & -2w & 0 & 0 \end{pmatrix}$$

(%i46) eigenvalues(A);

$$(\%o46) [[-iw, iw, 0], [1, 1, 2]]$$

(%i47) eigenvectors(A);

```
(%o47) [[[ -i w, i w, 0], [1, 1, 2]], [1, -i w, -2 i, -2 w], [1, i w, 2 i, -2 w], [0, 0, 1, 0]]
```

Esercizio 12. Risolvere il seguente sistema lineare mettendolo in forma matriciale.

$$\begin{cases} x - y + z = 1 \\ x + y + 2z = 2 \\ -3x + y - z = 1 \end{cases}$$

Esercizio 13. Data la matrice:

$$A = \begin{pmatrix} 2 & 1 & 1 \\ 1 & 2 & 1 \\ 1 & 1 & 2 \end{pmatrix}$$

calcolare il polinomio caratteristico $p(\gamma)$ e verificare che la matrice $p(A)$ è la matrice nulla.


10 Disegni in 2D e 3D

Maxima utilizza il programma gnuplot per rappresentare graficamente dei grafici. Di seguito vedremo la sintassi e l'uso dei due comandi più importanti per il disegno: `plot2d` e `plot3d`.

Il comando `plot2d` può essere molto utile per localizzare le radici di una equazione e per successivamente calcolarle con un metodo numerico, p.e. quello di Newton.

Esempio 17. Determinare le radici della seguente equazione: $\sin(x) - \log(x) = 0$. Evidentemente l'equazione non è risolvibile per via elementare. Facciamo il grafico della funzione $y = \sin(x) - \log(x)$ evidenziando gli assi cartesiani.

```
(%i48) y: sin(x)-log(x)$
(%i49) plot2d(y,[x,0,5],[gnuplot_preamble, "set zeroaxis"])$
(%i52)
```




Evidentemente la soluzione è “vicina” a 2.5, perciò useremo il metodo di Newton partendo da 2.5.

```
(%i52) load("newton")$
(%i53) newton(y,2.5);
Warning: Float to bigfloat conversion of 2.5
(%o53) 2.219107150437273B0
```

Esempio 18. Risolvere numericamente e graficamente il sistema:

$$\begin{cases} y = 2x - 3 \\ y = -x + 6 \end{cases}$$

```
(%i54) linsolve([alpha=2*beta-3, alpha=-beta+6],[alpha,beta]);
(%o58) [alpha=3, beta=3]
(%i59) plot2d([2*beta-3,-beta+6],[beta,0,5],[gnuplot_preamble, "set zeroaxis"])$
```




Evidentemente⁷² la coppia delle soluzioni è circa $(3, 3)$.

Maxima offre anche la possibilità di tracciare dei grafici in tre dimensioni, nota l'equazione cartesiana della superficie $\gamma = f(\alpha, \beta)$.

Esempio 19. Disegnare il grafico della funzione $\gamma = \cos(\sqrt{\alpha^2 + \beta})$

```
(%i60) plot3d(cos(sqrt(alpha^2+beta)), [alpha,-5,5], [beta,-5,5],
             [plot_format,gnuplot], [gnuplot_pm3d,true]);
```




Esempio 20. Trovare una approssimazione di $\cos(x)$ al secondo ordine intorno a zero e visualizzarne la bontà in intorni via via più grandi.

Utilizzeremo il comando `Taylor(f,x,0,2)` perché ci interessa lo sviluppo fino al secondo ordine.

```
(%i62) taylor(cos(x),x,0,2);
(%o62) 1 - x^2/2 + ...
(%i63) trunc(%); # Adesso diventa un polinomio
(%o63) 1 - x^2/2 + ...
(%i64) f:=%$
(%i70) errore(x):= (cos(x)-f)$
(%i88) x:0.01$
```

⁷². Del senno del poi son piene le fosse! :-))

```
(%i89) errore(x);
(%o89) 0.99995000041667 -
(%i90) x:0.1$
(%i91) errore(x);
(%o91) 0.99500416527803 -
(%i92) x:0.2
(%o96) 0.2
(%i97) errore(x);
(%o98) 0.98006657784124 -
(%i99) remvalue(x)$
(%i105) plot2d(cos(x)-(1-x^2/2),[x,-%pi/2,%pi/2],[gnuplot_preamble, "set
zeroaxis"])$
```



Concludendo fino a un angolo di circa 0.2 rad (cioè circa 12°) l'errore è circa del 2%.

Esercizio 14. Risolvere numericamente le seguenti equazioni:

- $\log(x^2 + 1) = \cos(x)$
- $e^{-x^2} = x^4 - 3$
- $e^{\sin(x)} = x + \frac{6}{5}$

11 Applicazioni

11.1 Urti elastici

Un carrello di massa m_1 e velocità v_1^i urta elasticamente un carrello di massa m_2 inizialmente fermo. Determinare le velocità dei carrelli dopo l'urto e discuterle supponendo $m_1 = m_2$, $m_1 \ll m_2$ e $m_1 \gg m_2$.

Siamo nella situazione di poter applicare la conservazione della quantità di moto e la conservazione dell'energia cinetica⁷³, indicando con v_{1f} e v_{2f} rispettivamente le velocità dopo l'urto del primo e del secondo carrello, possiamo scrivere:

$$m_1 v_{1i} = m_1 v_{1f} + m_2 v_{2f}$$

$$\frac{1}{2} m_1 v_{1i}^2 = \frac{1}{2} m_1 v_{1f}^2 + \frac{1}{2} m_2 v_{2f}^2$$

Risolvi le equazioni:

⁷³. Dato che i carrelli si muovono orizzontalmente, l'energia potenziale gravitazionale è costante e dato che l'energia meccanica si conserva segue l'asserto.

```
(%i1) eq1:m1*v1i=m1*v1f+m2*v2f$
(%i5) eq2:1/2*m1*v1i^2=1/2*m1*v1f^2+1/2*m2*v2f^2$
(%i6) algsys([eq1,eq2],[v1f,v2f]);
(%o6) [[v1f=v1i,v2f=0],[v1f=-((m2-m1)*v1i)/(m2+m1),v2f=2*m1*v1i/(m2+m1)]]
(%i7) v1:%o6[2][1]
(%o7) v1f=-((m2-m1)*v1i)/(m2+m1)
(%i8) v2:%o6[2][2]
(%o8) v2f=2*m1*v1i/(m2+m1)
```

Caso 1: $m_1 = m_2$.

```
(%i9) subst(m1,m2,v1)
(%o9) v1f=0
(%i10) subst(m1,m2,v2)
(%o10) v2f=v1i
```

Dunque dopo l'urto il primo carrello si ferma e il secondo procede con la stessa velocità del primo (plausibile).

Caso 2: $m_1 \ll m_2$. Per considerare questa situazione basterà calcolare il limite per m_2 che tende a più infinito.

```
(%i11) limit([v1,v2],m2,inf)
Is m1 v1i positive, negative, or zero?positive
(%o11) [v1f=-v1i,v2f=0]
```

In questa situazione m_2 è come un muro, non si muove e il primo carrello rimbalza (plausibile).

Caso 3: $m_1 \gg m_2$. Per considerare questa situazione basterà calcolare il limite per m_1 che tende a più infinito.

```
(%i12) limit([v1,v2],m1,inf)
(%o12) [v1f=v1i,v2f=2*v1i]
```

Quindi il primo carrello procede come se non fosse successo nulla (plausibile), mentre il secondo inizia a muoversi nella stessa direzione con velocità doppia (per niente plausibile).

Esercizio 15. Considerare due palloni di massa m_1 e m_2 posti a contatto uno sopra l'altro e fatti cadere per terra. Supponendo che tutti gli urti siano elastici, determinare le velocità dopo il rimbalzo e analizzare i casi $m_1 = m_2$, $m_1 \ll m_2$ e $m_1 \gg m_2$. Supporre il pavimento una massa illimitata...

11.2 Urti anelastici

Un carrello di massa m_1 e velocità v_{1i} urta un carrello di massa m_2 inizialmente fermo e vi resta attaccato. Determinare la velocità dei carrelli dopo l'urto e discutere la variazione della energia.

In virtù della conservazione della quantità di moto e del tipo di urto, possiamo scrivere:

$$m_1 v_{1i} = m_1 v_{1f} + m_2 v_{2f}$$

$$v_{1f} = v_{2f}$$

```
(%i1) eq1:m1*v1i=m1*v1f+m2*v2f$
(%i3) eq2:v1f=v2f$
```



```
(%i4) algsys([eq1,eq2],[v1f,v2f])
(%o5) [[v1f =  $\frac{m1 v1i}{m2 + m1}$ , v2f =  $\frac{m1 v1i}{m2 + m1}$ ]]
(%i6) part(%o5[1][1],2)
(%o30)  $\frac{m1 v1i}{m2 + m1}$ 
(%i31) vf:=%$
```

Calcoliamo le energie cinetiche prima e dopo l'urto.

```
(%i13) Ki:1/2*m1*v1i^2
(%o33)  $\frac{m1 v1i^2}{2}$ 
(%i34) Kf:1/2*(m1+m2)*vf^2
(%o35)  $\frac{m1^2 v1i^2}{2(m2 + m1)}$ 
(%i36) Kf/Ki
(%o36)  $\frac{m1}{m2 + m1}$ 
```

Possiamo concludere che c'è sempre perdita di energie e che è tanto maggiore quanto m_2 è più grande di m_1 , mentre è piccola se m_1 è molto più grande di m_2 .

11.3 Pallone gonfiato

Un pallone da basket ha la forma di una sfera. Quando ha inizialmente ha il volume di 10 l viene gonfiato con una pompa che trasferisce 3 l al minuto. Calcolare con quale velocità aumenta il suo raggio.

È noto che il volume di una sfera V è dato dalla formula:

$$V = \frac{4}{3}\pi r^3$$

la variabile “libera” è il tempo t , perciò deriviamo V rispetto al tempo:

```
(%i1) V(t):=4/3 * %pi * r(t)^3;
(%o1)  $V(t) := \frac{4}{3}\pi r(t)^3$ 
(%i2) diff(V(t),t);
(%o2)  $4\pi r(t)^2 \left(\frac{d}{dt} r(t)\right)$ 
```

Ora, dato che per $V = 3$ si ricava che $r = \sqrt[3]{\frac{3V}{4\pi}} = \sqrt[3]{\frac{3 \cdot 10}{4\pi}} = \sqrt[3]{\frac{15}{2\pi}}$. Perciò basta risolvere l'equazione:

$$10 = 4\pi \left(\sqrt[3]{\frac{15}{2\pi}} \right)^2 x$$

```
(%i3) solve(10=4*%pi*(15/(2*%pi))^(2/3)*x,x);
(%o3)  $\left[ x = \frac{5 \cdot 2^{\frac{2}{3}}}{2 \cdot 15^{\frac{2}{3}} \pi^{\frac{1}{3}}} \right]$ 
```

```
(%i4) float(%o3);
```

```
(%o5) [x = 0.44550153919066]
```

Concludendo il fattore di crescita del raggio è circa $x = 0.44$.

11.4 Studio di funzione

Studiare la seguente funzione algebrica razionale:

$$f(x) = \frac{x^2 - 1}{x^2 + 1}.$$

```
(%i1) f(x):=(x^2-1)/(x^2+1);
```

```
(%o1) f(x):=  $\frac{x^2 - 1}{x^2 + 1}$ 
```

```
(%i2) solve(f(x)=f(-x),x); # Verifico se è pari (sì!)
```

```
(%o2) all
```

```
(%i3) denom(f(x)); # Isolo il denominatore
```

```
(%o3)  $x^2 + 1$ 
```

```
(%i4) solve(%); # Dominio: tutti i reali
```

```
(%o4) [x = -i, x = i]
```

```
(%i5) limit(f(x),x,inf); # Limiti
```

```
(%o5) 1
```

```
(%i6) diff(f(x),x,1); # Derivata prima
```

```
(%o6)  $\frac{2x}{x^2 + 1} - \frac{2x(x^2 - 1)}{(x^2 + 1)^2}$ 
```

```
(%i7) ratsimp(%o6); # Semplifico la derivata prima
```

```
(%o10)  $\frac{4x}{x^4 + 2x^2 + 1}$ 
```

```
(%i11) solve(denom(%o10),x); # Dominio della derivata prima: ogni reale
```

```
(%o12) [x = -i, x = i]
```

```
(%i13) is(denom(%o6)>0); # Segno del denominatore della derivata
```

```
(%o13) true
```

```
(%i14) solve(num(%o10)=0,x); # Zero della derivata
```

```
(%o23) [x = 0]
```

```
(%i24) diff(f(x),x,2);
```

```
(%o24)  $\frac{2}{x^2 + 1} - \frac{8x^2}{(x^2 + 1)^2} - \frac{2(x^2 - 1)}{(x^2 + 1)^2} + \frac{8x^2(x^2 - 1)}{(x^2 + 1)^3}$ 
```

```
(%i25) ratsimp(%); # Derivata seconda semplificata
```

```
(%o25)  $-\frac{12x^2 - 4}{x^6 + 3x^4 + 3x^2 + 1}$ 
```

```
(%i26) is(denom(%o25)>0); # Vedi sopra
```

```
(%o26) true
```

```
(%i27) solve(num(%o25)=0,x); # Zeri della derivata seconda
```


$$(\%o27) \left[x = -\frac{1}{\sqrt{3}}, x = \frac{1}{\sqrt{3}} \right]$$

Concludendo:

- la funzione è pari (%o2),
- ha per dominio tutti i numeri reali (%o3-4),
- ha un asintoto orizzontale $y = 1$ (%o5),
- la derivata prima è positiva per $x > 0$ perciò ivi cresce la funzione (%o5-23),
- la derivata seconda è positiva nell'intervallo: $[-\frac{1}{\sqrt{3}}, +\frac{1}{\sqrt{3}}]$, perciò ivi ha concavità verso l'alto (%o24-27).

Ora il grafico che ottengo con il comando

```
plot2d(f(x),[x,-10,10],[gnuplot_preamble, "set zeroaxis;"]);
```



11.5 Sommatorie⁷⁶

Determinare il più grande valore di n per cui l'espressione numerica $\sum_{k=5}^n k$ non supera 10000.

I calcoli sono piuttosto facili, basta ricordare che $\sum_{k=1}^n k = \frac{n(n+1)}{2}$ e togliere 10 che è la somma dei primi quattro numeri ecc., ma con Maxima è ancora più facile.

```
(%i1) sum(k,k,5,n), simpsum;
```

$$(\%o1) \frac{n^2 + n}{2} - 10$$

```
(%i2) solve(%=10000,n);
```

$$(\%o2) \left[n = -\frac{\sqrt{80081} + 1}{2}, n = \frac{\sqrt{80081} - 1}{2} \right]$$

```
(%i3) float(%[2]);
```

$$(\%o3) n = 140.9929326856999$$

```
(%i4) sum(k,k,5,140);
```

⁷⁶. Questionario n. 2 Sessione ordinaria e suppletiva dell'Esame di Stato 2003-2004.

```
(%o4) 9860
(%i5) sum(k,k,5,141);
(%o5) 10001
```


Evidentemente la soluzione cercata è $n = 140$.

11.6 Confronto fra grafici⁷⁹

Risolvere la seguente disequazione in x : $(\ln x)^2 \geq \ln x^2$.

Facciamo un primo grafico (attenzione la scelta degli estremi non è casuale, ho fatto qualche prova) in modo da determinare gli intervalli solutivi e poi risolviamo l'equazione, è comunque ovvio che il dominio è $x > 0$.

```
(%i6) plot2d([(log(x))^2,log(x^2)], [x,0,10], [y,-20,20], [gnuplot_preamble,"set
zeroaxis;"])$
(%i7)
```



Notiamo tre intersezioni: una, α , intorno a uno e una, β , fra sette e otto. Calcoliamo α , β .

```
(%i7) solve((log(x))^2=log(x^2),x);
(%o1) [x = 1, x = e^2]
```

Dato che dobbiamo determinare gli intervalli per i quali $\log^2 x \geq \log x^2$ evidentemente le soluzioni sono: $]0, 1]$ e $[e^2, +\infty[$.

12 Esercizi riassuntivi

12.1 Insiemi

Esercizio 16. Di quanti elementi è composto l'insieme $\mathcal{P}(\mathcal{P}(\mathcal{P}(\phi)))$?

12.2 Algebra

Esercizio 17. Valutare la seguente espressione per $x=1$ e $y=1$.

$$\left(y - \frac{x^2 y}{y^2 - x^2}\right) \frac{y - x}{y^2 - 2x^2}.$$

⁷⁹. Questionario n. 4 Sessione ordinaria e suppletiva dell'Esame di Stato 2003-2004.

Esercizio 18. Riduci a forma normale il seguente monomio:

$$\left(-\frac{1}{3}\right)a^5b^2x(-5)a^{-2}bx^3y.$$

Esercizio 19. Semplifica le seguenti espressioni algebriche:

- a) $7x - 5x + 3x - 2x - 1$
- b) $2a^2x + 3a^2 - 7x + 4a^2 - a^2x - 7a^2$
- c) $6x + 3y - 9y + 2 + 5x - 2xy$
- d) $16ax^3y^2 : (-8xy^2)$
- e) $xy^3 : (-5ax^3y^2)$

Esercizio 20. Calcolare il MCD e il mcm fra i seguenti monomi:

$$2/3a^2b^2, a^3b^2x, 4/5a^2bx^3y$$

Esercizio 21. Calcola i seguenti prodotti notevoli:

- a) $(2a - b^2/3)^2$
- b) $(x + 2y - 1/2)^2$
- c) $(x - 1/2)^2$
- d) $(a + 3b - 1)(a + 3b + 1)$

Esercizio 22. Calcola quoziente e resto delle seguente divisione:

$$x^4 + 5x^3 + 5x^2 - 5x - 6 : x + 2$$

Esercizio 23. Fattorizza le seguenti espressioni:

- a) $16a^4 + b^4 - 8a^2b^2$
- b) $a^4 - 16b^4$
- c) $x^5 + 9xy^2 - 6x^3y$
- d) $x^2 - 9y^2 + 4x + 4$
- e) $a^2 + 9a + 8$
- f) $x^3 + 3x^2 - 10x$

Esercizio 24. Determina MCD e mcm fra i seguenti polinomi:

$$4 - 4x^2, 1 - x^3, 2x^2 - 4x + 2$$

Esercizio 25. Semplifica le seguenti frazioni algebriche:

- a) $\left(\frac{a}{a-1} - \frac{2}{1-a^2} \frac{1}{a+1}\right) : \frac{a^2 - a - 2}{a^2 - 2a + 1}$
- b) $\left(\frac{x+1}{2x} - \frac{x-1}{2x-1}\right) \frac{1-2x}{3x-1} - \left(1 - \frac{1}{x}\right) : \left(1 + \frac{1}{x-2}\right)$

Esercizio 26. Risolvi le seguenti equazioni fratte e intere:

- a) $\frac{2x}{4x^2-9} + \frac{3}{2x+3} = \frac{2}{x}$
- b) $3(a-1) + 2ax = a^2 - 1 - x(3-5a)$

Esercizio 27. In un rettangolo di perimetro 84cm la base è 3/4 dell'altezza. Calcola l'area del rettangolo.

Esercizio 28. Risolvi i seguenti sistemi lineari:

- a) $\begin{cases} 5x/4 + 2y/3 = 7 \\ 3x - 15 + y = 0 \end{cases}$
- b) $\begin{cases} 2x - 6y = 9 - 6z \\ x + y - 2z = 0 \\ 4x - 2z = 5 \end{cases}$

12.3 Analisi

Esercizio 29. Calcolare i seguenti limiti

- a) $\lim_{x \rightarrow 0} \frac{1 - \cos x}{x} [=0]$
- b) $\lim_{x \rightarrow 0} \frac{1 - \cos x}{x^2} [= \frac{1}{2}]$
- c) $\lim_{x \rightarrow 0} \frac{\tan x}{x} [=1]$

- d) $\lim_{x \rightarrow 0} \frac{\arcsin x}{x} [= 1]$
 e) $\lim_{x \rightarrow +\infty} (\sqrt{x^2+1} - x)\sqrt{x^2+2} [= \frac{1}{2}]$
 f) $\lim_{n \rightarrow +\infty} \left(1 + \frac{1}{n^2}\right)^n [= +\infty]$
 g) $\lim_{n \rightarrow +\infty} \left(1 + \frac{1}{n}\right)^{n^2} [= 1]$

Esercizio 30. Calcolare le derivate delle seguenti funzioni elementari (nella variabile x) semplificando fin dove possibile le espressioni ottenute.

- a) $D_x \frac{1-x^2}{1+x^2}$
 b) $D_x (1-2x)^3$
 c) $D_x \arcsin(1-x^2)$
 d) $D_x \sqrt{1+\sin^2 x}$
 e) $D_x \operatorname{atan} \frac{x}{a}$
 f) $D_x x^2 \ln x$
 g) $D_x \ln(ax)^{80}$
 h) $D_x \operatorname{atan} \frac{2x}{1-x^2}$
 i) $D_x \ln(x + \sqrt{1+x^2})$
 j) $D_x \sqrt{x}(1 + \cos x)$
 k) $D_x \ln \tan \frac{x}{2}$

Esercizio 31. Verificare che per $n=1, 2, 3, 4$ ⁸¹

$$D_x^n \left(\frac{1}{a-x} \right) = \frac{n!}{(a-x)^{n+1}}$$

$$\frac{e^x}{n!} D_x^n (e^{-x} x^n) = \sum_{k=0}^n \frac{n!}{(n-k)!k!} (-1)^k \frac{x^k}{k!} = L_n(x)$$

I polinomi $L_n(x)$ vengono detti polinomi di Laguerre.

Esercizio 32. Verificare mediante derivazione l'identità valida per $x > 0$:

$$\operatorname{atan} x + \operatorname{atan} \frac{1}{x} = \frac{\pi}{2}$$

Esercizio 33. Calcolare i seguenti integrali indefiniti:

- a) $\int x^2 e^x dx$
 b) $\int \frac{dx}{\sqrt{x}+1}$
 c) $\int \operatorname{atan} x dx$
 d) $\int \tan^2 x dx$
 e) $\int e^{ax} \cos bx dx$ (con a e b numeri reali)
 f) $\int \frac{dx}{x(1-x^3)}$

Esercizio 34. Verificare che se $|b| < a$ vale:

$$\int_0^\pi \frac{dx}{a+b \cos x} = \frac{\pi}{\sqrt{a^2-b^2}}$$

Esercizio 35. Calcolare i seguenti integrali generalizzati:

- a) $\int_0^1 \ln \frac{1}{x} dx$
 b) $\int_{-\infty}^{+\infty} \frac{dx}{1+x^4}$
 c) $\int_1^{+\infty} \frac{dx}{x^2 \ln x}$

80. Si noterà che la derivata non dipende dal parametro a . Come mai?

81. Ovviamente, visto che lavora Maxima, si può continuare.

Indice analitico

%	7	linsolve	14
addcol	18	load	6
addrow	19	matrix	13, 18
adjoiny	20	not	5
algsys	14	or	5
and	5	part	10
assume	15	partition	10
cardinality	3	partition_set	4
cartesian_product	3	plot3d	22
col	19	plot2d	12
determinant	19	powerset	3
diff	16	primep	6
eigenvalue	20	quotient	6
eigenvectors	21	ratsimp	9
expand	6	realpart	11
factor	6, 6	remainder	4, 6
float	5, 5, 14	romberg	18
fullratsimp	10	row	19
gcd	6	set	2
ident	18	setdifference	3
imagpart	11	solve	13
integrate	15, 18	sqrt	6
intersection	3	subset	4, 4
invert	19	subst	7
is	4	taylor	22
lcm	6	trunc	23
limit	15	union	3

Bibliografia

- [1] Enrico Centenaro. <http://www.centenaro.net>.
- [2] vari. <http://maxima.sf.net>.